
11.1 Jacobi Transformations of a Symmetric
Matrix

The Jacobi method consists of a sequence of orthogonal similarity transforma-
tions of the form of equation (11.0.14). Each transformation (a Jacobi rotation) is
just a plane rotation designed to annihilate one of the off-diagonal matrix elements.
Successive transformations undo previously set zeros, but the off-diagonal elements
nevertheless get smaller and smaller, until the matrix is diagonal to machine preci-
sion. Accumulating the product of the transformations as you go gives the matrix
of eigenvectors, equation (11.0.15), while the elements of the final diagonal matrix
are the eigenvalues.

The Jacobi method is absolutely foolproof for all real symmetric matrices. For
matrices of order greater than about 10, say, the algorithm is slower, by a significant
constant factor, than the QR method we shall give in §11.3. However, the Jacobi
algorithm is much simpler than the more efficient methods. We thus recommend it
for matrices of moderate order, where expense is not a major consideration.

The basic Jacobi rotation Ppq is a matrix of the form

Ppq =



1
· · ·

c · · · s
... 1

...
−s · · · c

· · ·
1


(11.1.1)

Here all the diagonal elements are unity except for the two elements c in rows (and
columns) p and q. All off-diagonal elements are zero except the two elements s and
−s. The numbers c and s are the cosine and sine of a rotation angle φ, so c2 +s2 = 1.

A plane rotation such as (11.1.1) is used to transform the matrix A according to

A′ = PTpq · A · Ppq (11.1.2)

Now, PTpq · A changes only rows p and q of A, while A · Ppq changes only columns
p and q. Notice that the subscripts p and q do not denote components of Ppq, but
rather label which kind of rotation the matrix is, i.e., which rows and columns it
affects. Thus the changed elements of A in (11.1.2) are only in the p and q rows
and columns indicated below:

A′ =



· · · a′1p · · · a′1q · · ·
...

...
...

...
a′p1 · · · a′pp · · · a′pq · · · a′pn

...
...

...
...

a′q1 · · · a′qp · · · a′qq · · · a′qn
...

...
...

...
· · · a′np · · · a′nq · · ·


(11.1.3)

Multiplying out equation (11.1.2) and using the symmetry of A, we get the explicit
formulas

a′rp = carp − sarq
a′rq = carq + sarp

r 6= p, r 6= q (11.1.4)

a′pp = c2app + s2aqq − 2scapq (11.1.5)

a′qq = s2app + c2aqq + 2scapq (11.1.6)

a′pq = (c2 − s2)apq + sc(app − aqq) (11.1.7)

The idea of the Jacobi method is to try to zero the off-diagonal elements by a
series of plane rotations. Accordingly, to set a′pq = 0, equation (11.1.7) gives the
following expression for the rotation angle φ

θ ≡ cot 2φ ≡ c2 − s2

2sc
=
aqq − app

2apq
(11.1.8)

If we let t ≡ s/c, the definition of θ can be rewritten

t2 + 2tθ− 1 = 0 (11.1.9)

The smaller root of this equation corresponds to a rotation angle less than π/4
in magnitude; this choice at each stage gives the most stable reduction. Using the
form of the quadratic formula with the discriminant in the denominator, we can
write this smaller root as

t =
sgn(θ)

|θ|+
√
θ2 + 1

(11.1.10)



458 Chapter 11. Eigensystems

W
orld W

ide W
eb sam

ple page from
 N

U
M

E
R

IC
A

L R
E

C
IP

E
S

 IN
 F

O
R

T
R

A
N

 77: T
H

E
 A

R
T

 O
F

 S
C

IE
N

T
IF

IC
 C

O
M

P
U

T
IN

G
 (IS

B
N

 0-521-43064-X
)

C
opyright (C

) 1988-1992 by C
am

bridge U
niversity P

ress.
P
rogram

s C
opyright (C

) 1988-1992 by N
um

erical R
ecipes S

oftw
are. 

P
erm

ission is granted for internet users to m
ake one paper copy for their ow

n personal use. F
urther reproduction, or any copying of m

achine-
readable files (including this one) to any server
com

puter, is strictly prohibited. T
o order N

um
erical R

ecipes books,
diskettes, or C
D

R
O

M
s

visit w
ebsite http://w

w
w

.nr.com
 or call 1-800-872-7423 (N

orth A
m

erica only),
or send em
ail to trade@

cup.cam
.ac.uk (outside N

orth A
m

erica).

If θ is so large that θ2 would overflow on the computer, we set t = 1/(2θ). It
now follows that

c =
1√
t2 + 1

(11.1.11)

s = tc (11.1.12)

When we actually use equations (11.1.4)–(11.1.7) numerically, we rewrite them
to minimize roundoff error. Equation (11.1.7) is replaced by

a′pq = 0 (11.1.13)

The idea in the remaining equations is to set the new quantity equal to the old
quantity plus a small correction. Thus we can use (11.1.7) and (11.1.13) to eliminate
aqq from (11.1.5), giving

a′pp = app − tapq (11.1.14)

Similarly,

a′qq = aqq + tapq (11.1.15)

a′rp = arp − s(arq + τarp) (11.1.16)

a′rq = arq + s(arp − τarq) (11.1.17)

where τ (= tanφ/2) is defined by

τ ≡ s

1 + c
(11.1.18)

One can see the convergence of the Jacobi method by considering the sum of
the squares of the off-diagonal elements

S =
∑
r 6=s
|ars|2 (11.1.19)

Equations (11.1.4)–(11.1.7) imply that

S′ = S − 2|apq|2 (11.1.20)

(Since the transformation is orthogonal, the sum of the squares of the diagonal
elements increases correspondingly by 2|apq|2.) The sequence of S’s thus decreases
monotonically. Since the sequence is bounded below by zero, and since we can
choose apq to be whatever element we want, the sequence can be made to converge
to zero.

Eventually one obtains a matrix D that is diagonal to machine precision. The
diagonal elements give the eigenvalues of the original matrix A, since

D = VT · A · V (11.1.21)

where
V = P1 · P2 · P3 · · · (11.1.22)

the Pi’s being the successive Jacobi rotation matrices. The columns of V are the
eigenvectors (since A · V = V · D). They can be computed by applying

V′ = V · Pi (11.1.23)

at each stage of calculation, where initially V is the identity matrix. In detail,
equation (11.1.23) is

v′rs = vrs (s 6= p, s 6= q)

v′rp = cvrp − svrq
v′rq = svrp + cvrq

(11.1.24)

We rewrite these equations in terms of τ as in equations (11.1.16) and (11.1.17)
to minimize roundoff.

The only remaining question is the strategy one should adopt for the order in
which the elements are to be annihilated. Jacobi’s original algorithmof 1846 searched
the whole upper triangle at each stage and set the largest off-diagonal element to zero.
This is a reasonable strategy for hand calculation, but it is prohibitive on a computer
since the search alone makes each Jacobi rotation a process of orderN2 instead ofN .

A better strategy for our purposes is the cyclic Jacobi method, where one
annihilates elements in strict order. For example, one can simply proceed down
the rows: P12,P13, ...,P1n; then P23,P24, etc. One can show that convergence
is generally quadratic for both the original or the cyclic Jacobi methods, for
nondegenerate eigenvalues. One such set of n(n − 1)/2 Jacobi rotations is called
a sweep.

The program below, based on the implementations in [1,2], uses two further
refinements:

• In the first three sweeps, we carry out the pq rotation only if |apq| > ε
for some threshold value

ε =
1

5

S0

n2
(11.1.25)

where S0 is the sum of the off-diagonal moduli,

S0 =
∑
r<s

|ars| (11.1.26)

• After four sweeps, if |apq| � |app| and |apq| � |aqq|, we set |apq| = 0
and skip the rotation. The criterion used in the comparison is |apq| <
10−(D+2)|app|, where D is the number of significant decimal digits on the
machine, and similarly for |aqq|.

In the following routine the n×n symmetric matrix a(1:n,1:n) is stored in
an np×np array. On output, the superdiagonal elements of a are destroyed, but the
diagonal and subdiagonal are unchanged and give full information on the original
symmetric matrix a. The parameter d is a vector of length np. On output, it returns
the eigenvalues of a in its first n elements. During the computation, it contains the
current diagonal of a. The matrix v outputs the normalized eigenvector belonging
to d(k) in its kth column. The parameter nrot is the number of Jacobi rotations
that were needed to achieve convergence.

Typical matrices require 6 to 10 sweeps to achieve convergence, or 3n2 to
5n2 Jacobi rotations. Each rotation requires of order 4n operations, each consisting
of a multiply and an add, so the total labor is of order 12n3 to 20n3 operations.
Calculation of the eigenvectors as well as the eigenvalues changes the operation
count from 4n to 6n per rotation, which is only a 50 percent overhead.

SUBROUTINE jacobi(a,n,np,d,v,nrot)
INTEGER n,np,nrot,NMAX
REAL a(np,np),d(np),v(np,np)
PARAMETER (NMAX=500)

Computes all eigenvalues and eigenvectors of a real symmetric matrix a, which is of size n
by n, stored in a physical np by np array. On output, elements of a above the diagonal are
destroyed. d returns the eigenvalues of a in its first n elements. v is a matrix with the same
logical and physical dimensions as a, whose columns contain, on output, the normalized
eigenvectors of a. nrot returns the number of Jacobi rotations that were required.

INTEGER i,ip,iq,j
REAL c,g,h,s,sm,t,tau,theta,tresh,b(NMAX),z(NMAX)
do 12 ip=1,n Initialize to the identity matrix.

do 11 iq=1,n
v(ip,iq)=0.

enddo 11

v(ip,ip)=1.
enddo 12

do 13 ip=1,n
b(ip)=a(ip,ip) Initialize b and d to the diagonal of a.
d(ip)=b(ip)
z(ip)=0. This vector will accumulate terms of the form tapq

as in equation (11.1.14).enddo 13

nrot=0
do 24 i=1,50

sm=0.
do 15 ip=1,n-1 Sum off-diagonal elements.

do 14 iq=ip+1,n
sm=sm+abs(a(ip,iq))

enddo 14

enddo 15

if(sm.eq.0.)return The normal return, which relies on quadratic conver-
gence to machine underflow.if(i.lt.4)then

tresh=0.2*sm/n**2 ...on the first three sweeps.
else

tresh=0. ...thereafter.
endif
do 22 ip=1,n-1

do 21 iq=ip+1,n
g=100.*abs(a(ip,iq))

After four sweeps, skip the rotation if the off-diagonal element is small.
if((i.gt.4).and.(abs(d(ip))+g.eq.abs(d(ip)))

* .and.(abs(d(iq))+g.eq.abs(d(iq))))then
a(ip,iq)=0.

else if(abs(a(ip,iq)).gt.tresh)then
h=d(iq)-d(ip)
if(abs(h)+g.eq.abs(h))then

t=a(ip,iq)/h t = 1/(2θ)
else

theta=0.5*h/a(ip,iq) Equation (11.1.10).
t=1./(abs(theta)+sqrt(1.+theta**2))
if(theta.lt.0.)t=-t

endif
c=1./sqrt(1+t**2)
s=t*c
tau=s/(1.+c)
h=t*a(ip,iq)
z(ip)=z(ip)-h
z(iq)=z(iq)+h
d(ip)=d(ip)-h
d(iq)=d(iq)+h
a(ip,iq)=0.
do 16 j=1,ip-1 Case of rotations 1 ≤ j < p.

g=a(j,ip)
h=a(j,iq)
a(j,ip)=g-s*(h+g*tau)
a(j,iq)=h+s*(g-h*tau)

enddo 16

do 17 j=ip+1,iq-1 Case of rotations p < j < q.
g=a(ip,j)
h=a(j,iq)
a(ip,j)=g-s*(h+g*tau)
a(j,iq)=h+s*(g-h*tau)

enddo 17

do 18 j=iq+1,n Case of rotations q < j ≤ n.
g=a(ip,j)
h=a(iq,j)
a(ip,j)=g-s*(h+g*tau)
a(iq,j)=h+s*(g-h*tau)

enddo 18

do 19 j=1,n
g=v(j,ip)
h=v(j,iq)
v(j,ip)=g-s*(h+g*tau)
v(j,iq)=h+s*(g-h*tau)

enddo 19

nrot=nrot+1
endif

enddo 21

enddo 22

do 23 ip=1,n
b(ip)=b(ip)+z(ip)
d(ip)=b(ip) Update d with the sum of tapq,
z(ip)=0. and reinitialize z.

enddo 23

enddo 24

pause ’too many iterations in jacobi’
return
END

Note that the above routine assumes that underflows are set to zero. On
machines where this is not true, the program must be modified.

The eigenvalues are not ordered on output. If sorting is desired, the following
routine can be invoked to reorder the output of jacobi or of later routines in this
chapter. (The method, straight insertion, is N2 rather than N logN ; but since you
have just done an N3 procedure to get the eigenvalues, you can afford yourself
this little indulgence.)

SUBROUTINE eigsrt(d,v,n,np)
INTEGER n,np
REAL d(np),v(np,np)

Given the eigenvalues d and eigenvectors v as output from jacobi (§11.1) or tqli (§11.3),
this routine sorts the eigenvalues into descending order, and rearranges the columns of v
correspondingly. The method is straight insertion.

INTEGER i,j,k
REAL p
do 13 i=1,n-1

k=i
p=d(i)
do 11 j=i+1,n

if(d(j).ge.p)then
k=j
p=d(j)

endif
enddo 11

if(k.ne.i)then
d(k)=d(i)
d(i)=p
do 12 j=1,n

p=v(j,i)
v(j,i)=v(j,k)
v(j,k)=p

enddo 12

endif
enddo 13

return
END

11.2 Reduction of a Symmetric Matrix
to Tridiagonal Form: Givens and
Householder Reductions

As already mentioned, the optimum strategy for finding eigenvalues and
eigenvectors is, first, to reduce the matrix to a simple form, only then beginning an
iterative procedure. For symmetric matrices, the preferred simple form is tridiagonal.
The Givens reduction is a modification of the Jacobi method. Instead of trying to
reduce the matrix all the way to diagonal form, we are content to stop when the
matrix is tridiagonal. This allows the procedure to be carried out in a finite number
of steps, unlike the Jacobi method, which requires iteration to convergence.


